// File: contracts/Factory.sol



// SPDX-License-Identifier: MIT
pragma solidity =0.8.4;
pragma abicoder v2;

import "@openzeppelin/contracts/utils/Address.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/utils/cryptography/ECDSA.sol";

import "./interfaces/ISignatureValidator.sol";
import "./interfaces/external/INonfungiblePositionManager.sol";
import "./interfaces/external/ISwapRouter02.sol";
import "./interfaces/external/IUniswapV3Pool.sol";

import "./interfaces/IFindNFT.sol";
import "./interfaces/IMortgage.sol";
import "./interfaces/IMortgagePoolFactory.sol";
import "./interfaces/IMath.sol";

import "./libraries/BytesLib.sol";
import "./libraries/TransferHelper.sol";
import "./libraries/UniswapCreatePoolHelper.sol";
import "./libraries/UniswapConstants.sol";

import "./OSP.sol";

contract Factory is Ownable {
  using BytesLib for bytes;
  using ECDSA for bytes32;

  struct FindInfo {
    address token;
    address pool;
    uint256 cnftTokenId;
    uint256 onftTokenId;
    uint256[] lpTokenIdList;
    uint24 fee;
  }

  struct OspInfo {
    uint256 poolConfigIndex;
    uint256 stars;
    address pool;
    uint256 cnftTokenId;
    uint256 onftTokenId;
    uint256[] lpTokenIdList;
    string projectId;
  }

  struct OspPoolConfigPool {
    UniswapStruct.Position[] positions;
    uint160 initSqrtPriceX96;
  }

  struct OspPoolConfigParams {
    uint24 fee;
    OspPoolConfigPool ospFindPool;
    OspPoolConfigPool findOspPool;
  }

  struct OspPoolConfig {
    uint24 fee;
    OspPoolConfigPool ospFindPool;
    OspPoolConfigPool findOspPool;
    uint256 totalSupply;
  }

  struct CreateOSPBaseParams {
    string name;
    string symbol;
    string projectId;
    uint256 stars;
    uint256 poolConfigIndex;
    uint256 nftPercentConfigIndex;
  }

  struct CreateOSPByProjectOwnerParams {
    CreateOSPBaseParams base;
    uint256 deadline;
    bytes signature;
  }

  struct CreateOSPParams {
    CreateOSPBaseParams base;
    uint256 deadline;
    uint256 buyNFTTokenAmountMax;
    uint256 buyNFTFindAmount;
    bytes tokenToFindOutPath;
    bytes signature;
  }

  struct NFTPercent {
    uint24 cnft;
    uint24 onft;
  }

  address public immutable weth;
  address public immutable earn;
  address public immutable findnft;
  address public immutable mortgage;
  address public immutable mortgagePoolFactory;
  address public immutable math;

  address public signatureAddress;
  bool public disableSetSignatureAddressFlag;

  FindInfo public findInfo;

  NFTPercent[] public nftPercentConfigs;
  mapping(uint256 => OspPoolConfig) private ospPoolConfigs;
  uint256 public ospPoolConfigsCount;

  mapping(string => address) public projectId2OspToken;
  mapping(address => OspInfo) public token2OspInfo;

  event SetSignatureAddress(address signatureAddress);
  event DisableSetSignatureAddress();
  event AddOspPoolConfig(OspPoolConfigParams config);
  event AddNFTPercentConfig(uint24 cnftPercent, uint24 onftPercent);
  event CreateOSPByProjectOwner(CreateOSPByProjectOwnerParams params);
  event CreateOSPByProjectOwnerAndMultiply(
    CreateOSPByProjectOwnerParams params,
    uint256 tokenId
  );
  event CreateOSP(CreateOSPParams params);
  event CreateOSPAndMultiply(CreateOSPParams params, uint256 tokenId);

  constructor(
    address _find,
    address _weth,
    address _earn,
    address _findnft,
    address _mortgage,
    address _mortgagePoolFactory,
    address _math,
    address _signatureAddress
  ) {
    findInfo.fee = 100;

    findInfo.token = _find;
    weth = _weth;
    earn = _earn;
    findnft = _findnft;
    mortgage = _mortgage;
    mortgagePoolFactory = _mortgagePoolFactory;
    math = _math;
    signatureAddress = _signatureAddress;
  }

  modifier checkDeadline(uint256 deadline) {
    require(block.timestamp <= deadline, "CDL");
    _;
  }

  function setSignatureAddress(address _signatureAddress) external onlyOwner {
    require(!disableSetSignatureAddressFlag, "DE");

    signatureAddress = _signatureAddress;
    emit SetSignatureAddress(_signatureAddress);
  }

  function disableSetSignatureAddress() external onlyOwner {
    disableSetSignatureAddressFlag = true;
    emit DisableSetSignatureAddress();
  }

  function addOspPoolConfig(OspPoolConfigParams memory config)
    external
    onlyOwner
  {
    uint256 index = ospPoolConfigsCount;
    ospPoolConfigsCount = ospPoolConfigsCount + 1;

    ospPoolConfigs[index].fee = config.fee;
    ospPoolConfigs[index].ospFindPool.initSqrtPriceX96 = config
      .ospFindPool
      .initSqrtPriceX96;
    ospPoolConfigs[index].findOspPool.initSqrtPriceX96 = config
      .findOspPool
      .initSqrtPriceX96;
    for (uint256 i = 0; i < config.ospFindPool.positions.length; i++) {
      ospPoolConfigs[index].ospFindPool.positions.push(
        config.ospFindPool.positions[i]
      );
    }
    for (uint256 i = 0; i < config.findOspPool.positions.length; i++) {
      ospPoolConfigs[index].findOspPool.positions.push(
        config.findOspPool.positions[i]
      );
    }

    ospPoolConfigs[index].totalSupply = _getTotalSupplyFromParams(config);

    (
      address leftPool,
      uint256[] memory leftTokenIdList,
      address rightPool,
      uint256[] memory rightTokenIdList
    ) = IMortgagePoolFactory(mortgagePoolFactory).createPool(index);

    IMath(math).createPoolInfo(index, leftPool, leftTokenIdList, false);
    IMath(math).createPoolInfo(index, rightPool, rightTokenIdList, true);

    emit AddOspPoolConfig(config);
  }

  function addNFTPercentConfig(uint24 _cnftPercent, uint24 _onftPercent)
    external
    onlyOwner
  {
    require(_cnftPercent + _onftPercent == 10000, "E");

    nftPercentConfigs.push(
      NFTPercent({ cnft: _cnftPercent, onft: _onftPercent })
    );
    emit AddNFTPercentConfig(_cnftPercent, _onftPercent);
  }

  function getOspPoolConfigs(uint256 index)
    external
    view
    returns (
      uint24 fee,
      uint160 findOspPoolInitSqrtPriceX96,
      UniswapStruct.Position[] memory findOspPoolPositions,
      uint160 ospFindPoolInitSqrtPriceX96,
      UniswapStruct.Position[] memory ospFindPoolPositions,
      uint256 totalSupply
    )
  {
    OspPoolConfig memory config = ospPoolConfigs[index];
    fee = config.fee;

    findOspPoolInitSqrtPriceX96 = config.findOspPool.initSqrtPriceX96;
    findOspPoolPositions = config.findOspPool.positions;

    ospFindPoolInitSqrtPriceX96 = config.ospFindPool.initSqrtPriceX96;
    ospFindPoolPositions = config.ospFindPool.positions;

    totalSupply = config.totalSupply;
  }

  function findLpTokenIdList()
    external
    view
    returns (uint256[] memory lpTokenIdList)
  {
    lpTokenIdList = findInfo.lpTokenIdList;
  }

  function ospLpTokenIdList(address osp)
    external
    view
    returns (uint256[] memory lpTokenIdList)
  {
    lpTokenIdList = token2OspInfo[osp].lpTokenIdList;
  }

  function createFindUniswapPool() external onlyOwner {
    require(earn != address(0));
    // prevent duplicate creation
    require(findInfo.pool == address(0));

    (address pool, uint256[] memory lpTokenIdList) = _createFindUniswapPool();
    (uint256 cnftTokenId, uint256 onftTokenId) = _createFindNFT();

    findInfo.pool = pool;
    findInfo.cnftTokenId = cnftTokenId;
    findInfo.onftTokenId = onftTokenId;
    findInfo.lpTokenIdList = lpTokenIdList;
  }

  function createOSPByProjectOwner(CreateOSPByProjectOwnerParams memory params)
    external
    checkDeadline(params.deadline)
  {
    _verifyCreateByProjectOwnerSignature(params);

    _createOSPWithoutPay(params.base);

    emit CreateOSPByProjectOwner(params);
  }

  function createOSPByProjectOwnerAndMultiply(
    CreateOSPByProjectOwnerParams memory params,
    bytes memory tokenToFindOutPath,
    uint256 AllFindAmount,
    uint256 amountPayMax
  )
    external
    payable
    checkDeadline(params.deadline)
    returns (
      uint256 positionOspAmountDelta,
      uint256 amountNeedPay,
      address tokenPay,
      uint256 tokenId
    )
  {
    _verifyCreateByProjectOwnerSignature(params);

    address osp = _createOSPWithoutPay(params.base);

    address tokenIn = tokenToFindOutPath.toAddress(
      tokenToFindOutPath.length - 20
    );
    _processSenderTransfer(tokenIn, amountPayMax);

    _approveToTarget(tokenIn, mortgage, amountPayMax);

    (positionOspAmountDelta, amountNeedPay, tokenPay, tokenId) = _multiply(
      osp,
      AllFindAmount,
      amountPayMax,
      tokenToFindOutPath
    );

    _refund(tokenIn, amountPayMax - amountNeedPay);

    emit CreateOSPByProjectOwnerAndMultiply(params, tokenId);
  }

  function createOSP(CreateOSPParams memory params)
    external
    payable
    checkDeadline(params.deadline)
    returns (uint256 nftNeedPay, address tokenPay)
  {
    _verifyCreateSignature(params);

    _createOSPWithoutPay(params.base);

    address tokenIn = params.tokenToFindOutPath.toAddress(
      params.tokenToFindOutPath.length - 20
    );

    if (params.buyNFTFindAmount == 0) {
      require(params.buyNFTTokenAmountMax == 0, "E1");
      require(msg.value == 0, "E2");
      tokenPay = tokenIn;
    } else {
      _processSenderTransfer(tokenIn, params.buyNFTTokenAmountMax);

      _approveToTarget(
        tokenIn,
        UniswapConstants.UNISWAP_ROUTER,
        params.buyNFTTokenAmountMax
      );

      (tokenPay, nftNeedPay) = _pathSwapToFindAndTransferToEarn(
        params.tokenToFindOutPath,
        params.buyNFTTokenAmountMax,
        params.buyNFTFindAmount
      );

      _refund(tokenIn, params.buyNFTTokenAmountMax - nftNeedPay);
    }

    emit CreateOSP(params);
  }

  function createOSPAndMultiply(
    CreateOSPParams memory params,
    uint256 AllFindAmount,
    uint256 amountPayMax
  )
    external
    payable
    checkDeadline(params.deadline)
    returns (
      uint256 nftNeedPay,
      uint256 positionOspAmountDelta,
      uint256 amountNeedPay,
      address tokenPay,
      uint256 tokenId
    )
  {
    _verifyCreateSignature(params);

    address osp = _createOSPWithoutPay(params.base);

    address tokenIn = params.tokenToFindOutPath.toAddress(
      params.tokenToFindOutPath.length - 20
    );

    _processSenderTransfer(tokenIn, params.buyNFTTokenAmountMax + amountPayMax);

    if (params.buyNFTFindAmount == 0) {
      require(params.buyNFTTokenAmountMax == 0, "E1");
    } else {
      _approveToTarget(
        tokenIn,
        UniswapConstants.UNISWAP_ROUTER,
        params.buyNFTTokenAmountMax
      );

      (, nftNeedPay) = _pathSwapToFindAndTransferToEarn(
        params.tokenToFindOutPath,
        params.buyNFTTokenAmountMax,
        params.buyNFTFindAmount
      );
    }

    _approveToTarget(tokenIn, mortgage, amountPayMax);

    (positionOspAmountDelta, amountNeedPay, tokenPay, tokenId) = _multiply(
      osp,
      AllFindAmount,
      amountPayMax,
      params.tokenToFindOutPath
    );

    _refund(
      tokenIn,
      params.buyNFTTokenAmountMax + amountPayMax - nftNeedPay - amountNeedPay
    );

    emit CreateOSPAndMultiply(params, tokenId);
  }

  function _multiply(
    address osp,
    uint256 AllFindAmount,
    uint256 amountPayMax,
    bytes memory tokenToFindOutPath
  )
    private
    returns (
      uint256 positionOspAmountDelta,
      uint256 amountNeedPay,
      address tokenPay,
      uint256 tokenId
    )
  {
    uint256 value = msg.value == 0 ? 0 : amountPayMax;
    (positionOspAmountDelta, , amountNeedPay, tokenPay, tokenId) = IMortgage(
      mortgage
    ).multiply{ value: value }(
      osp,
      AllFindAmount,
      amountPayMax,
      tokenToFindOutPath
    );
    // transferFrom position nft
    IMortgage(mortgage).safeTransferFrom(address(this), _msgSender(), tokenId);
  }

  function _refund(address token, uint256 amount) private {
    if (amount > 0) {
      if (token != _weth9()) {
        TransferHelper.safeTransfer(token, _msgSender(), amount);
      } else {
        TransferHelper.safeTransferETH(_msgSender(), amount);
      }
    }
  }

  function _createOSPWithoutPay(CreateOSPBaseParams memory params)
    private
    returns (address osp)
  {
    require(earn != address(0));
    require(findInfo.token != address(0));
    require(projectId2OspToken[params.projectId] == address(0), "E");

    osp = address(
      new OSP(
        address(this),
        params.name,
        params.symbol,
        ospPoolConfigs[params.poolConfigIndex].totalSupply
      )
    );

    (address pool, uint256[] memory lpTokenIdList) = _createOspUniswapPool(
      osp,
      params
    );
    (uint256 cnftTokenId, uint256 onftTokenId) = _createOspNFT(
      osp,
      params,
      _msgSender(),
      earn
    );

    token2OspInfo[osp].poolConfigIndex = params.poolConfigIndex;
    token2OspInfo[osp].projectId = params.projectId;
    token2OspInfo[osp].stars = params.stars;
    token2OspInfo[osp].pool = pool;
    token2OspInfo[osp].cnftTokenId = cnftTokenId;
    token2OspInfo[osp].onftTokenId = onftTokenId;
    token2OspInfo[osp].lpTokenIdList = lpTokenIdList;

    projectId2OspToken[params.projectId] = osp;
  }

  function _getTotalSupplyFromParams(OspPoolConfigParams memory config)
    private
    pure
    returns (uint256 totalSupply)
  {
    uint256 ospFindTotalSupply;
    uint256 findOspTotalSupply;
    for (
      uint256 index = 0;
      index < config.ospFindPool.positions.length;
      index++
    ) {
      ospFindTotalSupply += config.ospFindPool.positions[index].amount;
    }
    for (
      uint256 index = 0;
      index < config.findOspPool.positions.length;
      index++
    ) {
      findOspTotalSupply += config.findOspPool.positions[index].amount;
    }
    require(findOspTotalSupply == ospFindTotalSupply, "TSE");
    totalSupply = findOspTotalSupply;
  }

  function _createOspUniswapPool(address osp, CreateOSPBaseParams memory params)
    private
    returns (address pool, uint256[] memory lpTokenIdList)
  {
    OspPoolConfig memory config = ospPoolConfigs[params.poolConfigIndex];
    require(config.fee != 0);
    OspPoolConfigPool memory poolParams;
    if (osp < findInfo.token) {
      poolParams = config.ospFindPool;
    } else {
      poolParams = config.findOspPool;
    }

    (pool, lpTokenIdList) = UniswapCreatePoolHelper.createUniswapPool(
      UniswapCreatePoolHelper.PoolParams({
        baseToken: findInfo.token,
        newToken: osp,
        fee: config.fee,
        initSqrtPriceX96: poolParams.initSqrtPriceX96,
        recipient: earn,
        positions: poolParams.positions
      })
    );
  }

  function _createFindUniswapPool()
    private
    returns (address pool, uint256[] memory lpTokenIdList)
  {
    (
      uint160 initSqrtPriceX96,
      UniswapStruct.Position[] memory positions
    ) = _getFindEthPoolPrams();
    (pool, lpTokenIdList) = UniswapCreatePoolHelper.createUniswapPool(
      UniswapCreatePoolHelper.PoolParams({
        baseToken: weth,
        newToken: findInfo.token,
        fee: findInfo.fee,
        initSqrtPriceX96: initSqrtPriceX96,
        recipient: earn,
        positions: positions
      })
    );
  }

  function _getFindEthPoolPrams()
    private
    view
    returns (
      uint160 initSqrtPriceX96,
      UniswapStruct.Position[] memory positions
    )
  {
    positions = new UniswapStruct.Position[](1);
    if (weth > findInfo.token) {
      // 0 is find, 1 is weth
      // int(math.sqrt(0.001 * (2**192)))
      initSqrtPriceX96 = 2505289222420813103173471269;
      positions[0] = UniswapStruct.Position({
        tickLower: -69082,
        tickUpper: -69081,
        amount: IERC20(findInfo.token).totalSupply()
      });
    } else {
      // 1 is find, 0 is weth
      // int(math.sqrt(1000 * (2**192)))
      initSqrtPriceX96 = 2505539751343055434542618388703;
      positions[0] = UniswapStruct.Position({
        tickLower: 69080,
        tickUpper: 69081,
        amount: IERC20(findInfo.token).totalSupply()
      });
    }
  }

  function _createOspNFT(
    address osp,
    CreateOSPBaseParams memory params,
    address cnftOwner,
    address onftOwner
  ) private returns (uint256 cnftTokenId, uint256 onftTokenId) {
    NFTPercent memory config = nftPercentConfigs[params.nftPercentConfigIndex];
    cnftTokenId = IFindNFT(findnft).mint(
      IFindNFT.MintParams({
        name: params.name,
        symbol: params.symbol,
        projectId: params.projectId,
        stars: params.stars,
        token: osp,
        percent: config.cnft,
        isCnft: true,
        owner: cnftOwner
      })
    );
    onftTokenId = IFindNFT(findnft).mint(
      IFindNFT.MintParams({
        name: params.name,
        symbol: params.symbol,
        projectId: params.projectId,
        stars: params.stars,
        token: osp,
        percent: config.onft,
        isCnft: false,
        owner: onftOwner
      })
    );
  }

  function _createFindNFT()
    private
    returns (uint256 cnftTokenId, uint256 onftTokenId)
  {
    NFTPercent memory config = nftPercentConfigs[0];

    IFindNFT.MintParams memory params = IFindNFT.MintParams({
      name: "github.com/0xfind",
      symbol: "0xHARBERGER",
      projectId: "github/105404818/000000",
      stars: 1,
      token: findInfo.token,
      percent: config.cnft,
      isCnft: true,
      owner: owner()
    });

    cnftTokenId = IFindNFT(findnft).mint(params);

    params.percent = config.onft;
    params.isCnft = false;
    onftTokenId = IFindNFT(findnft).mint(params);
  }

  function _processSenderTransfer(address tokenIn, uint256 amount) private {
    if (tokenIn != _weth9()) {
      require(msg.value == 0, "MVE1");
      TransferHelper.safeTransferFrom(
        tokenIn,
        _msgSender(),
        address(this),
        amount
      );
    } else {
      require(msg.value == amount, "MVE2");
    }
  }

  function _approveToTarget(
    address tokenIn,
    address target,
    uint256 amount
  ) private {
    if (tokenIn != _weth9()) {
      if (IERC20(tokenIn).allowance(address(this), target) < amount) {
        IERC20(tokenIn).approve(target, type(uint256).max);
      }
    }
  }

  function _pathSwapToFindAndTransferToEarn(
    bytes memory tokenToFindOutPath,
    uint256 buyNFTTokenAmountMax,
    uint256 buyNFTFindAmount
  ) private returns (address tokenIn, uint256 amountIn) {
    address tokenOut = tokenToFindOutPath.toAddress(0);
    tokenIn = tokenToFindOutPath.toAddress(tokenToFindOutPath.length - 20);

    require(tokenOut == findInfo.token, "PE");

    if (tokenIn == tokenOut) {
      require(tokenToFindOutPath.length == 20, "PE1");

      tokenIn = findInfo.token;
      amountIn = buyNFTFindAmount;
      TransferHelper.safeTransfer(findInfo.token, earn, amountIn);
    } else {
      // cannot contain multiply token
      uint256 value = msg.value == 0 ? 0 : buyNFTTokenAmountMax;
      amountIn = ISwapRouter02(UniswapConstants.UNISWAP_ROUTER).exactOutput{
        value: value
      }(
        ISwapRouter02.ExactOutputParams({
          path: tokenToFindOutPath,
          recipient: earn,
          amountOut: buyNFTFindAmount,
          amountInMaximum: buyNFTTokenAmountMax
        })
      );

      ISwapRouter02(UniswapConstants.UNISWAP_ROUTER).refundETH();
    }
    require(buyNFTTokenAmountMax >= amountIn, "ME");
  }

  function _weth9() private view returns (address) {
    return
      INonfungiblePositionManager(UniswapConstants.UNISWAP_V3_POSITIONS)
        .WETH9();
  }

  function _verifyCreateByProjectOwnerSignature(
    CreateOSPByProjectOwnerParams memory params
  ) private view {
    bytes32 raw = keccak256(
      abi.encode(params.base, params.deadline, _msgSender())
    );

    _verifySignature(raw, params.signature);
  }

  function _verifyCreateSignature(CreateOSPParams memory params) private view {
    bytes32 raw = keccak256(
      abi.encode(
        params.base,
        params.deadline,
        params.buyNFTTokenAmountMax,
        params.buyNFTFindAmount,
        params.tokenToFindOutPath,
        _msgSender()
      )
    );

    _verifySignature(raw, params.signature);
  }

  function _verifySignature(bytes32 raw, bytes memory signature) private view {
    if (Address.isContract(signatureAddress)) {
      require(signature.length == 0, "SLE");
      require(
        ISignatureValidator(signatureAddress).isValidHash(
          raw.toEthSignedMessageHash()
        ),
        "SE1"
      );
    } else {
      require(
        raw.toEthSignedMessageHash().recover(signature) == signatureAddress,
        "SE2"
      );
    }
  }

  receive() external payable {}
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly
                /// @solidity memory-safe-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/utils/cryptography/ECDSA.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/cryptography/ECDSA.sol)

pragma solidity ^0.8.0;

import "../Strings.sol";

/**
 * @dev Elliptic Curve Digital Signature Algorithm (ECDSA) operations.
 *
 * These functions can be used to verify that a message was signed by the holder
 * of the private keys of a given address.
 */
library ECDSA {
    enum RecoverError {
        NoError,
        InvalidSignature,
        InvalidSignatureLength,
        InvalidSignatureS,
        InvalidSignatureV
    }

    function _throwError(RecoverError error) private pure {
        if (error == RecoverError.NoError) {
            return; // no error: do nothing
        } else if (error == RecoverError.InvalidSignature) {
            revert("ECDSA: invalid signature");
        } else if (error == RecoverError.InvalidSignatureLength) {
            revert("ECDSA: invalid signature length");
        } else if (error == RecoverError.InvalidSignatureS) {
            revert("ECDSA: invalid signature 's' value");
        } else if (error == RecoverError.InvalidSignatureV) {
            revert("ECDSA: invalid signature 'v' value");
        }
    }

    /**
     * @dev Returns the address that signed a hashed message (`hash`) with
     * `signature` or error string. This address can then be used for verification purposes.
     *
     * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
     * this function rejects them by requiring the `s` value to be in the lower
     * half order, and the `v` value to be either 27 or 28.
     *
     * IMPORTANT: `hash` _must_ be the result of a hash operation for the
     * verification to be secure: it is possible to craft signatures that
     * recover to arbitrary addresses for non-hashed data. A safe way to ensure
     * this is by receiving a hash of the original message (which may otherwise
     * be too long), and then calling {toEthSignedMessageHash} on it.
     *
     * Documentation for signature generation:
     * - with https://web3js.readthedocs.io/en/v1.3.4/web3-eth-accounts.html#sign[Web3.js]
     * - with https://docs.ethers.io/v5/api/signer/#Signer-signMessage[ethers]
     *
     * _Available since v4.3._
     */
    function tryRecover(bytes32 hash, bytes memory signature) internal pure returns (address, RecoverError) {
        // Check the signature length
        // - case 65: r,s,v signature (standard)
        // - case 64: r,vs signature (cf https://eips.ethereum.org/EIPS/eip-2098) _Available since v4.1._
        if (signature.length == 65) {
            bytes32 r;
            bytes32 s;
            uint8 v;
            // ecrecover takes the signature parameters, and the only way to get them
            // currently is to use assembly.
            /// @solidity memory-safe-assembly
            assembly {
                r := mload(add(signature, 0x20))
                s := mload(add(signature, 0x40))
                v := byte(0, mload(add(signature, 0x60)))
            }
            return tryRecover(hash, v, r, s);
        } else if (signature.length == 64) {
            bytes32 r;
            bytes32 vs;
            // ecrecover takes the signature parameters, and the only way to get them
            // currently is to use assembly.
            /// @solidity memory-safe-assembly
            assembly {
                r := mload(add(signature, 0x20))
                vs := mload(add(signature, 0x40))
            }
            return tryRecover(hash, r, vs);
        } else {
            return (address(0), RecoverError.InvalidSignatureLength);
        }
    }

    /**
     * @dev Returns the address that signed a hashed message (`hash`) with
     * `signature`. This address can then be used for verification purposes.
     *
     * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
     * this function rejects them by requiring the `s` value to be in the lower
     * half order, and the `v` value to be either 27 or 28.
     *
     * IMPORTANT: `hash` _must_ be the result of a hash operation for the
     * verification to be secure: it is possible to craft signatures that
     * recover to arbitrary addresses for non-hashed data. A safe way to ensure
     * this is by receiving a hash of the original message (which may otherwise
     * be too long), and then calling {toEthSignedMessageHash} on it.
     */
    function recover(bytes32 hash, bytes memory signature) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, signature);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Overload of {ECDSA-tryRecover} that receives the `r` and `vs` short-signature fields separately.
     *
     * See https://eips.ethereum.org/EIPS/eip-2098[EIP-2098 short signatures]
     *
     * _Available since v4.3._
     */
    function tryRecover(
        bytes32 hash,
        bytes32 r,
        bytes32 vs
    ) internal pure returns (address, RecoverError) {
        bytes32 s = vs & bytes32(0x7fffffffffffffffffffffffffffffffffffffffffffffffffffffffffffffff);
        uint8 v = uint8((uint256(vs) >> 255) + 27);
        return tryRecover(hash, v, r, s);
    }

    /**
     * @dev Overload of {ECDSA-recover} that receives the `r and `vs` short-signature fields separately.
     *
     * _Available since v4.2._
     */
    function recover(
        bytes32 hash,
        bytes32 r,
        bytes32 vs
    ) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, r, vs);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Overload of {ECDSA-tryRecover} that receives the `v`,
     * `r` and `s` signature fields separately.
     *
     * _Available since v4.3._
     */
    function tryRecover(
        bytes32 hash,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal pure returns (address, RecoverError) {
        // EIP-2 still allows signature malleability for ecrecover(). Remove this possibility and make the signature
        // unique. Appendix F in the Ethereum Yellow paper (https://ethereum.github.io/yellowpaper/paper.pdf), defines
        // the valid range for s in (301): 0 < s < secp256k1n ÷ 2 + 1, and for v in (302): v ∈ {27, 28}. Most
        // signatures from current libraries generate a unique signature with an s-value in the lower half order.
        //
        // If your library generates malleable signatures, such as s-values in the upper range, calculate a new s-value
        // with 0xFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFEBAAEDCE6AF48A03BBFD25E8CD0364141 - s1 and flip v from 27 to 28 or
        // vice versa. If your library also generates signatures with 0/1 for v instead 27/28, add 27 to v to accept
        // these malleable signatures as well.
        if (uint256(s) > 0x7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF5D576E7357A4501DDFE92F46681B20A0) {
            return (address(0), RecoverError.InvalidSignatureS);
        }
        if (v != 27 && v != 28) {
            return (address(0), RecoverError.InvalidSignatureV);
        }

        // If the signature is valid (and not malleable), return the signer address
        address signer = ecrecover(hash, v, r, s);
        if (signer == address(0)) {
            return (address(0), RecoverError.InvalidSignature);
        }

        return (signer, RecoverError.NoError);
    }

    /**
     * @dev Overload of {ECDSA-recover} that receives the `v`,
     * `r` and `s` signature fields separately.
     */
    function recover(
        bytes32 hash,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, v, r, s);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Returns an Ethereum Signed Message, created from a `hash`. This
     * produces hash corresponding to the one signed with the
     * https://eth.wiki/json-rpc/API#eth_sign[`eth_sign`]
     * JSON-RPC method as part of EIP-191.
     *
     * See {recover}.
     */
    function toEthSignedMessageHash(bytes32 hash) internal pure returns (bytes32) {
        // 32 is the length in bytes of hash,
        // enforced by the type signature above
        return keccak256(abi.encodePacked("\x19Ethereum Signed Message:\n32", hash));
    }

    /**
     * @dev Returns an Ethereum Signed Message, created from `s`. This
     * produces hash corresponding to the one signed with the
     * https://eth.wiki/json-rpc/API#eth_sign[`eth_sign`]
     * JSON-RPC method as part of EIP-191.
     *
     * See {recover}.
     */
    function toEthSignedMessageHash(bytes memory s) internal pure returns (bytes32) {
        return keccak256(abi.encodePacked("\x19Ethereum Signed Message:\n", Strings.toString(s.length), s));
    }

    /**
     * @dev Returns an Ethereum Signed Typed Data, created from a
     * `domainSeparator` and a `structHash`. This produces hash corresponding
     * to the one signed with the
     * https://eips.ethereum.org/EIPS/eip-712[`eth_signTypedData`]
     * JSON-RPC method as part of EIP-712.
     *
     * See {recover}.
     */
    function toTypedDataHash(bytes32 domainSeparator, bytes32 structHash) internal pure returns (bytes32) {
        return keccak256(abi.encodePacked("\x19\x01", domainSeparator, structHash));
    }
}

// File: contracts/interfaces/ISignatureValidator.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.4;

interface ISignatureValidator {
  function isValidSignature(bytes32 _hash, bytes memory _signature)
    external
    view
    returns (bytes4 magicValue);

  function isValidHash(bytes32 _hash) external view returns (bool isValid);
}

// File: contracts/interfaces/external/INonfungiblePositionManager.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.7.6;
pragma abicoder v2;

interface INonfungiblePositionManager {
  function tokenOfOwnerByIndex(address owner, uint256 index)
    external
    view
    returns (uint256 tokenId);

  function refundETH() external payable;

  function factory() external view returns (address);

  function WETH9() external view returns (address);

  function balanceOf(address owner) external view returns (uint256 balance);

  function ownerOf(uint256 tokenId) external view returns (address owner);

  function positions(uint256 tokenId)
    external
    view
    returns (
      uint96 nonce,
      address operator,
      address token0,
      address token1,
      uint24 fee,
      int24 tickLower,
      int24 tickUpper,
      uint128 liquidity,
      uint256 feeGrowthInside0LastX128,
      uint256 feeGrowthInside1LastX128,
      uint128 tokensOwed0,
      uint128 tokensOwed1
    );

  struct MintParams {
    address token0;
    address token1;
    uint24 fee;
    int24 tickLower;
    int24 tickUpper;
    uint256 amount0Desired;
    uint256 amount1Desired;
    uint256 amount0Min;
    uint256 amount1Min;
    address recipient;
    uint256 deadline;
  }

  function createAndInitializePoolIfNecessary(
    address token0,
    address token1,
    uint24 fee,
    uint160 sqrtPriceX96
  ) external payable returns (address pool);

  function mint(MintParams calldata params)
    external
    payable
    returns (
      uint256 tokenId,
      uint128 liquidity,
      uint256 amount0,
      uint256 amount1
    );

  struct CollectParams {
    uint256 tokenId;
    address recipient;
    uint128 amount0Max;
    uint128 amount1Max;
  }

  function collect(CollectParams calldata params)
    external
    payable
    returns (uint256 amount0, uint256 amount1);
}

// File: contracts/interfaces/external/ISwapRouter02.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.4;

interface ISwapRouter02 {
  struct ExactInputSingleParams {
    address tokenIn;
    address tokenOut;
    uint24 fee;
    address recipient;
    uint256 amountIn;
    uint256 amountOutMinimum;
    uint160 sqrtPriceLimitX96;
  }

  function exactInputSingle(ExactInputSingleParams calldata params)
    external
    payable
    returns (uint256 amountOut);

  struct ExactInputParams {
    bytes path;
    address recipient;
    uint256 amountIn;
    uint256 amountOutMinimum;
  }

  function exactInput(ExactInputParams calldata params)
    external
    payable
    returns (uint256 amountOut);

  struct ExactOutputSingleParams {
    address tokenIn;
    address tokenOut;
    uint24 fee;
    address recipient;
    uint256 amountOut;
    uint256 amountInMaximum;
    uint160 sqrtPriceLimitX96;
  }

  function exactOutputSingle(ExactOutputSingleParams calldata params)
    external
    payable
    returns (uint256 amountIn);

  struct ExactOutputParams {
    bytes path;
    address recipient;
    uint256 amountOut;
    uint256 amountInMaximum;
  }

  function exactOutput(ExactOutputParams calldata params)
    external
    payable
    returns (uint256 amountIn);

  function refundETH() external payable;
}

// File: contracts/interfaces/external/IUniswapV3Pool.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.7.6;

interface IUniswapV3Pool {
  function token0() external view returns (address);

  function token1() external view returns (address);

  function fee() external view returns (uint24);

  function slot0()
    external
    view
    returns (
      uint160 sqrtPriceX96,
      int24 tick,
      uint16 observationIndex,
      uint16 observationCardinality,
      uint16 observationCardinalityNext,
      uint8 feeProtocol,
      bool unlocked
    );

  function tickSpacing() external view returns (int24);

  function maxLiquidityPerTick() external view returns (uint128);

  function liquidity() external view returns (uint128);

  function ticks(int24 tick)
    external
    view
    returns (
      uint128 liquidityGross,
      int128 liquidityNet,
      uint256 feeGrowthOutside0X128,
      uint256 feeGrowthOutside1X128,
      int56 tickCumulativeOutside,
      uint160 secondsPerLiquidityOutsideX128,
      uint32 secondsOutside,
      bool initialized
    );
}

// File: contracts/interfaces/IFindNFT.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.4;

interface IFindNFT {
  struct Info {
    string name;
    string symbol;
    string projectId;
    uint256 stars;
    address token;
    uint256 percent;
    bool isCnft;
    uint256 tokenId;
    uint256 timestamp;
  }

  struct MintParams {
    string name;
    string symbol;
    string projectId;
    uint256 stars;
    address token;
    uint256 percent;
    bool isCnft;
    address owner;
  }

  function ownerOf(uint256 tokenId) external view returns (address owner);

  function tokenId2Info(uint256 _tokenId)
    external
    view
    returns (
      string memory name,
      string memory symbol,
      string memory projectId,
      uint256 stars,
      address token,
      uint256 percent,
      bool isCnft,
      uint256 tokenId,
      uint256 timestamp
    );

  function mint(MintParams memory params) external returns (uint256 tokenId);

  function safeTransferFrom(
    address from,
    address to,
    uint256 tokenId
  ) external;

  function isClaimed(uint256 tokenId) external view returns (bool);

  function claim(uint256 tokenId) external;
}

// File: contracts/interfaces/IMortgage.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.4;

interface IMortgage {
  function multiply(
    address ospAsset,
    uint256 AllFindAmount,
    uint256 amountPayMax,
    bytes memory otherToFindOutPath
  )
    external
    payable
    returns (
      uint256 positionOspAmountDelta,
      uint256 payFindAmount,
      uint256 amountNeedPay,
      address tokenPay,
      uint256 tokenId
    );

  function safeTransferFrom(
    address from,
    address to,
    uint256 tokenId
  ) external;

  function positions(uint256 _tokenId)
    external
    view
    returns (
      uint256 tokenId,
      address ospAsset,
      uint256 ospAmount
    );
}

// File: contracts/interfaces/IMortgagePoolFactory.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.4;

interface IMortgagePoolFactory {
  function createPool(uint256 poolConfigIndex)
    external
    returns (
      address leftPool,
      uint256[] memory leftTokenIdList,
      address rightPool,
      uint256[] memory rightTokenIdList
    );
}

// File: contracts/interfaces/IMath.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.4;

interface IMath {
  function createPoolInfo(
    uint256 poolConfigIndex,
    address pool,
    uint256[] memory tokenIdList,
    bool isFindOspPool
  ) external;

  function mortgage(address osp, uint256 ospAmount)
    external
    returns (uint256 findAmount);
}

// File: contracts/libraries/BytesLib.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.4;

library BytesLib {
  function slice(
    bytes memory _bytes,
    uint256 _start,
    uint256 _length
  ) internal pure returns (bytes memory) {
    require(_length + 31 >= _length, "slice_overflow");
    require(_start + _length >= _start, "slice_overflow");
    require(_bytes.length >= _start + _length, "slice_outOfBounds");

    bytes memory tempBytes;

    assembly {
      switch iszero(_length)
      case 0 {
        tempBytes := mload(0x40)

        let lengthmod := and(_length, 31)

        let mc := add(add(tempBytes, lengthmod), mul(0x20, iszero(lengthmod)))
        let end := add(mc, _length)

        for {
          let cc := add(
            add(add(_bytes, lengthmod), mul(0x20, iszero(lengthmod))),
            _start
          )
        } lt(mc, end) {
          mc := add(mc, 0x20)
          cc := add(cc, 0x20)
        } {
          mstore(mc, mload(cc))
        }

        mstore(tempBytes, _length)

        mstore(0x40, and(add(mc, 31), not(31)))
      }
      default {
        tempBytes := mload(0x40)

        mstore(tempBytes, 0)

        mstore(0x40, add(tempBytes, 0x20))
      }
    }

    return tempBytes;
  }

  function toAddress(bytes memory _bytes, uint256 _start)
    internal
    pure
    returns (address)
  {
    require(_start + 20 >= _start, "toAddress_overflow");
    require(_bytes.length >= _start + 20, "toAddress_outOfBounds");
    address tempAddress;

    assembly {
      tempAddress := div(
        mload(add(add(_bytes, 0x20), _start)),
        0x1000000000000000000000000
      )
    }

    return tempAddress;
  }

  function toUint24(bytes memory _bytes, uint256 _start)
    internal
    pure
    returns (uint24)
  {
    require(_start + 3 >= _start, "toUint24_overflow");
    require(_bytes.length >= _start + 3, "toUint24_outOfBounds");
    uint24 tempUint;

    assembly {
      tempUint := mload(add(add(_bytes, 0x3), _start))
    }

    return tempUint;
  }
}

// File: contracts/libraries/TransferHelper.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.4;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

library TransferHelper {
  function safeTransfer(
    address token,
    address to,
    uint256 value
  ) internal {
    (bool success, bytes memory data) = token.call(
      abi.encodeWithSelector(IERC20.transfer.selector, to, value)
    );
    require(success && (data.length == 0 || abi.decode(data, (bool))), "ST");
  }

  function safeTransferFrom(
    address token,
    address from,
    address to,
    uint256 value
  ) internal {
    (bool success, bytes memory data) = token.call(
      abi.encodeWithSelector(IERC20.transferFrom.selector, from, to, value)
    );
    require(success && (data.length == 0 || abi.decode(data, (bool))), "STF");
  }

  function safeTransferETH(address to, uint256 value) internal {
    (bool success, ) = to.call{ value: value }(new bytes(0));
    require(success, "STE");
  }
}

// File: contracts/libraries/UniswapCreatePoolHelper.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.4;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

import "../interfaces/external/INonfungiblePositionManager.sol";
import "./UniswapConstants.sol";
import "./UniswapStruct.sol";

library UniswapCreatePoolHelper {
  struct PoolParams {
    address baseToken;
    address newToken;
    uint24 fee;
    uint160 initSqrtPriceX96;
    address recipient;
    UniswapStruct.Position[] positions;
  }

  function createUniswapPool(PoolParams memory params)
    internal
    returns (address pool, uint256[] memory tokenIdList)
  {
    address token0;
    address token1;
    {
      if (params.baseToken < params.newToken) {
        token0 = params.baseToken;
        token1 = params.newToken;
      } else {
        token0 = params.newToken;
        token1 = params.baseToken;
      }
    }

    pool = INonfungiblePositionManager(UniswapConstants.UNISWAP_V3_POSITIONS)
      .createAndInitializePoolIfNecessary(
        token0,
        token1,
        params.fee,
        params.initSqrtPriceX96
      );

    IERC20(params.newToken).approve(
      UniswapConstants.UNISWAP_V3_POSITIONS,
      type(uint256).max
    );

    uint256 amount0Desired;
    uint256 amount1Desired;
    tokenIdList = new uint256[](params.positions.length);
    for (uint256 index = 0; index < params.positions.length; index++) {
      if (token1 == params.newToken) {
        amount0Desired = 0;
        amount1Desired = params.positions[index].amount;
      }
      if (token0 == params.newToken) {
        amount0Desired = params.positions[index].amount;
        amount1Desired = 0;
      }

      INonfungiblePositionManager.MintParams memory mintParams;
      {
        mintParams = INonfungiblePositionManager.MintParams({
          token0: token0,
          token1: token1,
          fee: params.fee,
          tickLower: params.positions[index].tickLower,
          tickUpper: params.positions[index].tickUpper,
          amount0Desired: amount0Desired,
          amount1Desired: amount1Desired,
          amount0Min: 0,
          amount1Min: 0,
          recipient: params.recipient,
          deadline: block.timestamp
        });
      }

      (uint256 tokenId, , , ) = INonfungiblePositionManager(
        UniswapConstants.UNISWAP_V3_POSITIONS
      ).mint(mintParams);
      tokenIdList[index] = tokenId;
    }
  }
}

// File: contracts/libraries/UniswapConstants.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.7.6;

library UniswapConstants {
  address public constant UNISWAP_V3_POSITIONS =
    0xC36442b4a4522E871399CD717aBDD847Ab11FE88;
  address public constant UNISWAP_ROUTER =
    0x68b3465833fb72A70ecDF485E0e4C7bD8665Fc45;
}

// File: contracts/OSP.sol



// SPDX-License-Identifier: MIT
pragma solidity =0.8.4;

import "@openzeppelin/contracts/token/ERC20/ERC20.sol";

contract OSP is ERC20 {
  constructor(
    address factory,
    string memory name,
    string memory symbol,
    uint256 _totalSupply
  ) ERC20(name, symbol) {
    _mint(factory, _totalSupply);
  }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: contracts/libraries/UniswapStruct.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.7.6;

library UniswapStruct {
  struct Position {
    int24 tickLower;
    int24 tickUpper;
    uint256 amount;
  }
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * The default value of {decimals} is 18. To select a different value for
     * {decimals} you should overload it.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless this function is
     * overridden;
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
        }
        _balances[to] += amount;

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        _balances[account] += amount;
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
        }
        _totalSupply -= amount;

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}